NAG Library Routine Document

F08HEF (DSBTRD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08HEF (DSBTRD) reduces a real symmetric band matrix to tridiagonal form.

2 Specification

```
SUBROUTINE F08HEF (VECT, UPLO, N, KD, AB, LDAB, D, E, Q, LDQ, WORK,        &
                   INFO)
INTEGER             N, KD, LDAB, LDQ, INFO
REAL (KIND=nag_wp) AB(LDAB,*), D(N), E(N-1), Q(LDQ,*), WORK(N)
CHARACTER(1)        VECT, UPLO
```

The routine may be called by its LAPACK name ***dsbtrd***.

3 Description

F08HEF (DSBTRD) reduces a symmetric band matrix A to symmetric tridiagonal form T by an orthogonal similarity transformation:

$$T = Q^{\mathrm{T}} A Q.$$

The orthogonal matrix Q is determined as a product of Givens rotation matrices, and may be formed explicitly by the routine if required.

The routine uses a vectorizable form of the reduction, due to Kaufman (1984).

4 References

Kaufman L (1984) Banded eigenvalue solvers on vector machines *ACM Trans. Math. Software* **10** 73–86

Parlett B N (1998) *The Symmetric Eigenvalue Problem* SIAM, Philadelphia

5 Parameters

1: VECT – CHARACTER(1) *Input*

On entry: indicates whether Q is to be returned.

VECT = 'V'
Q is returned.

VECT = 'U'
Q is updated (and the array Q must contain a matrix on entry).

VECT = 'N'
Q is not required.

Constraint: VECT = 'V', 'U' or 'N'.

2: UPLO – CHARACTER(1) *Input*

On entry: indicates whether the upper or lower triangular part of A is stored.

UPLO = 'U'
The upper triangular part of A is stored.

UPLO = 'L'
The lower triangular part of A is stored.

Constraint: UPLO = 'U' or 'L'.

3: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

4: KD – INTEGER *Input*

On entry: if UPLO = 'U', the number of superdiagonals, k_d, of the matrix A.

If UPLO = 'L', the number of subdiagonals, k_d, of the matrix A.

Constraint: KD ≥ 0.

5: AB(LDAB, $*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array AB must be at least $\max(1, \text{N})$.

On entry: the upper or lower triangle of the n by n symmetric band matrix A.

The matrix is stored in rows 1 to $k_d + 1$, more precisely,

if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element A_{ij} in $\text{AB}(k_d + 1 + i - j, j)$ for $\max(1, j - k_d) \leq i \leq j$;

if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element A_{ij} in $\text{AB}(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_d)$.

On exit: AB is overwritten by values generated during the reduction to tridiagonal form.

The first superdiagonal or subdiagonal and the diagonal of the tridiagonal matrix T are returned in AB using the same storage format as described above.

6: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F08HEF (DSBTRD) is called.

Constraint: LDAB $\geq \max(1, \text{KD} + 1)$.

7: D(N) – REAL (KIND=nag_wp) array *Output*

On exit: the diagonal elements of the tridiagonal matrix T.

8: E(N $-$ 1) – REAL (KIND=nag_wp) array *Output*

On exit: the off-diagonal elements of the tridiagonal matrix T.

9: Q(LDQ, $*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array Q must be at least $\max(1, \text{N})$ if VECT = 'V' or 'U' and at least 1 if VECT = 'N'.

On entry: if VECT = 'U', Q must contain the matrix formed in a previous stage of the reduction (for example, the reduction of a banded symmetric-definite generalized eigenproblem); otherwise Q need not be set.

On exit: if VECT = 'V' or 'U', the n by n matrix Q.

If VECT = 'N', Q is not referenced.

10: LDQ – INTEGER *Input*

On entry: the first dimension of the array Q as declared in the (sub)program from which F08HEF (DSBTRD) is called.

Constraints:

if VECT = 'V' or 'U', LDQ $\geq \max(1, \mathrm{N})$;
if VECT = 'N', LDQ ≥ 1.

11: WORK(N) – REAL (KIND=nag_wp) array *Workspace*

12: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed tridiagonal matrix T is exactly similar to a nearby matrix $(A + E)$, where

$$\|E\|_2 \leq c(n)\epsilon\|A\|_2,$$

$c(n)$ is a modestly increasing function of n, and ϵ is the ***machine precision***.

The elements of T themselves may be sensitive to small perturbations in A or to rounding errors in the computation, but this does not affect the stability of the eigenvalues and eigenvectors.

The computed matrix Q differs from an exactly orthogonal matrix by a matrix E such that

$$\|E\|_2 = O(\epsilon),$$

where ϵ is the ***machine precision***.

8 Parallelism and Performance

F08HEF (DSBTRD) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08HEF (DSBTRD) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately $6n^2k$ if VECT = 'N' with $3n^3(k-1)/k$ additional operations if VECT = 'V'.

The complex analogue of this routine is F08HSF (ZHBTRD).

10 Example

This example computes all the eigenvalues and eigenvectors of the matrix A, where

$$A = \begin{pmatrix} 4.99 & 0.04 & 0.22 & 0.00 \\ 0.04 & 1.05 & -0.79 & 1.04 \\ 0.22 & -0.79 & -2.31 & -1.30 \\ 0.00 & 1.04 & -1.30 & -0.43 \end{pmatrix}.$$

Here A is symmetric and is treated as a band matrix. The program first calls F08HEF (DSBTRD) to reduce A to tridiagonal form T, and to form the orthogonal matrix Q; the results are then passed to F08JEF (DSTEQR) which computes the eigenvalues and eigenvectors of A.

10.1 Program Text

```
    Program f08hefe

!     F08HEF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
      Use nag_library, Only: dsbtrd, dsteqr, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, j, kd, ldab, ldq, n
      Character (1)                    :: uplo
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: ab(:,:), d(:), e(:), q(:,:), work(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: max, min
!     .. Executable Statements ..
      Write (nout,*) 'F08HEF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kd
      ldab = kd + 1
      ldq = n
      Allocate (ab(ldab,n),d(n),e(n-1),q(ldq,n),work(2*n-2))

!     Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Do i = 1, n
          Read (nin,*)(ab(kd+1+i-j,j),j=i,min(n,i+kd))
        End Do
      Else If (uplo=='L') Then
        Do i = 1, n
          Read (nin,*)(ab(1+i-j,j),j=max(1,i-kd),i)
        End Do
      End If

!     Reduce A to tridiagonal form T = (Q**T)*A*Q (and form Q)
!     The NAG name equivalent of dsbtrd is f08hef
      Call dsbtrd('V',uplo,n,kd,ab,ldab,d,e,q,ldq,work,info)

!     Calculate all the eigenvalues and eigenvectors of A
!     The NAG name equivalent of dsteqr is f08jef
      Call dsteqr('V',n,d,e,q,ldq,work,info)

      Write (nout,*)
      If (info>0) Then
        Write (nout,*) 'Failure to converge.'
      Else
```

```
!       Print eigenvalues and eigenvectors

        Write (nout,*) 'Eigenvalues'
        Write (nout,99999) d(1:n)
        Write (nout,*)
        Flush (nout)

!       Standardize the eigenvectors so that first elements are non-negative.
        Do i = 1, n
          If (q(1,i)<0.0_nag_wp) q(1:n,i) = -q(1:n,i)
        End Do

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04caf('General',' ',n,n,q,ldq,'Eigenvectors',ifail)

      End If

99999 Format (3X,(8F8.4))
    End Program f08hefe
```

10.2 Program Data

```
F08HEF Example Program Data
  4  2                        :Values of N and KD
  'L'                         :Value of UPLO
  4.99
  0.04   1.05
  0.22  -0.79  -2.31
         1.04  -1.30  -0.43   :End of matrix A
```

10.3 Program Results

```
 F08HEF Example Program Results

 Eigenvalues
    -2.9943 -0.7000  1.9974  4.9969

 Eigenvectors
          1       2       3       4
 1   0.0251  0.0162  0.0113  0.9995
 2  -0.0656 -0.5859  0.8077  0.0020
 3  -0.9002 -0.3135 -0.3006  0.0311
 4  -0.4298  0.7471  0.5070 -0.0071
```